
Computing the area and winding number
for a Bézier curve
Boguslaw Jackowski

Introduction

Why would we want to compute the area or winding
number for a given (closed) Bézier curve? The general answer is: in computational graphics, various
measures of graphical objects may prove useful.
For example, MetaPost was equipped with a
very important function, missing from Metafont:
arclength, which computes the length of a given
arc. A typical problem that can be easily solved
using this function is placing uniformly spaced text
along a curve.
In my font application, I needed a function that
calculates a distance between two curves — this feature could be used to compute an approximation of
a multi-node Bézier curve by a single Bézier arc (i.e.,
for simplifying curves). Another operation I needed
is a Boolean function telling whether a given curve
is embedded in another curve.
These operations are missing from both Metafont and MetaPost, although they are feasible in
Metafont due to its bitmap operations.
A useful distance between curves a and b can be
computed as the number of pixels that receive a nonzero value when filling the curve a--b--cycle (for
a given resolution). Also, checking two closed curves
a and b for mutual embedding could be calculated
(again, for a given resolution) by filling the curve a
and unfilling the curve b — if no pixels with a positive
value remain, it means that a is embedded in b.
It should be emphasized that the result depends
on the resolution, which can certainly be considered
a drawback.
In MetaPost, bitmap operations are unavailable, hence computing the distance between curves
and checking the mutual embedding of curves cannot be done simply. However, a distance between
two non-intersecting curves, a and b, can be computed as the absolute value of the area surrounded
by a--b--cycle; if the curves intersect, one has to
find all intersections and compute the area for all
subcycles, which is a fairly complex task (due to
numerical instability).
On the other hand, a winding number — for nontouching curves — can be used for determining the
mutual position of the two curves (as explained in
the section below).
While bitmap operations are practically insensitive to the tangency of curves, the algorithm for
computing the winding number presented here cru-

cially is. Nevertheless, these MetaPost algorithms
for computing the area and winding number may be
considered counterparts of the relevant bitmap-based
Metafont algorithms. Of course, I would still gladly
welcome building bitmap operations into MetaPost.
2

Area enclosed by a cyclic Bézier spline

The area between the graph of a function x 7→
(x, C(x)) and the x-axis is shown as the hatched
region in this figure:
y
y1
y = C(x)

y0
x0

x1

x

It can be computed as the integral
Z x1
C(x)dx

(1)

x0

If the curve is given parametrically, i.e., t 7→ (C x (t),
C y (t)), the integral (1) can be rewritten by substituting x = C x (t), x0 = C x (t0 ), x1 = C x (t1 ),

x
C(x) = C(C x (t)) = C y (t), and dx = dCdt(t) dt as
Z t1
dC x (t)
C y (t)
dt
(2)
dt
t0
Furthermore, if t0 6= t1 and
(Cx (t0 ), Cy (t0 )) = (Cx (t1 ), Cy (t1 )),
i.e., the curve is cyclic, the integral (2) yields the
area surrounded by the curve.
Assume that the cyclic curve is a spline composed of Bézier arcs B1 , B2 , . . . , Bn (each defined
for 0 ≤ t ≤ 1). The area of the region surrounded by
the spline shown here:
y

B3 (t)

B2 (t)

B1 (t)

x

is the sum of integrals:
n Z 1
X
dB x (t)
Biy (t) i
dt
dt
i=1 0

In the following, I’ll skip the index i, as calculations
are exactly the same for each i; the functions B(t) =
(B x (t), B y (t)) are third-degree polynomials:
B(t) = b0 (1 − t)3 + 3b1 (1 − t)2 t + 3b2 (1 − t) t2 + b3 t3
where b0 = (bx0 , by0 ), b1 = (bx1 , by1 ), b2 = (bx2 , by2 ), b3 =
(bx3 , by3 ) are points in the plane; b0 , b3 are the nodes
and b1 , b2 are the control points of the Bézier arc B.
The computation of the antiderivative of the
x
function B y (t) dBdt(t) (a fifth-degree polynomial) is
an elementary task (actually, it suffices to know that
a derivative of tn is ntn−1 and, thus, the integral of
1 n+1
tn is n+1
t
). Skipping tedious calculations, I’ll
present the final formula:
Z 1
dB x (t)
20 B y (t)
dt = (bx1 −bx0 )(10by0 +6by1 +3by2 +by3 )
dt
0
+(bx2 − bx1 )(4by0 + 6by1 + 6by2 + 4by3 )
+(bx3 − bx2 )(by0 + 3by1 + 6by2 + 10by3 )
This formula stemmed from the discussion between Daniel Luecking and Laurent Siebenmann on
the Metafont/MetaPost discussion list (metafont@
ens.fr, 2000; presently the MetaPost discussion
list is hosted by TUG — http://lists.tug.org/
metapost). Luecking made a crucial observation that
three real multiplications per Bézier arc sufficient
to compute the area surrounded by a Bézier spline;
division of the whole sum by 20 is a constant cost
and thus can be neglected. Integer multiplication
can be replaced by operations usually faster than real
multiplication (e.g., 10a = 8a + 2a, 8a = a shifted
left by 3 bits, 2a = a shifted left by 1 bit).
Of course, such an optimization of the arithmetic operations makes sense only in a “production”
implementation of the algorithm. The implementation at the level of Metafont/MetaPost macros can
be neither efficient nor precise. Nevertheless, the
following code may sometimes prove useful:
% p is a B\’ezier segment; result = \int y dx
vardef area(expr p) =
save xa, xb, xc, xd, ya, yb, yc, yd;
(xa,20ya) = point 0 of p;
(xb,20yb) = postcontrol 0 of p;
(xc,20yc) = precontrol 1 of p;
(xd,20yd) = point 1 of p;
(xb-xa)*(10ya + 6yb + 3yc +
yd)
+(xc-xb)*( 4ya + 6yb + 6yc + 4yd)
+(xd-xc)*( ya + 3yb + 6yc + 10yd)
enddef;
% P is a cyclic path; result = area of interior
vardef Area(expr P) =
area(subpath (0,1) of P)
for t=1 upto length(P)-1:
+ area(subpath (t,t+1) of P) endfor
enddef;

Observe that the macro Area computes a signed
area: negative for counterclockwise-oriented curves,
and positive for clockwise-oriented ones. As a consequence, a non-trivial curve with self-intersection(s)
(e.g., eight-shaped) may surround a region with the
area equal to zero.
Observe also that the calculations can be carried
out with respect to the y-axis, thus the following code
% p is a B\’ezier segment; result = \int y dx
vardef area(expr p) =
save xa, xb, xc, xd, ya, yb, yc, yd;
(-20xa,ya) = point 0 of p;
(-20xb,yb) = postcontrol 0 of p;
(-20xc,yc) = precontrol 1 of p;
(-20xd,yd) = point 1 of p;
(yb-ya)*(10xa + 6xb + 3xc +
xd)
+(yc-yb)*( 4xa + 6xb + 6xc + 4xd)
+(yd-yc)*( xa + 3xb + 6xc + 10xd)
enddef;
% P is a cyclic path; result = area of interior
vardef Area(expr P) =
area(subpath (0,1) of P)
for t=1 upto length(P)-1:
+ area(subpath (t,t+1) of P) endfor
enddef;

will yield the same results as the former (within the
accuracy of rounding errors).

A winding number for Bézier splines

Assume that we are given a point P in the plane
and the planar curve C(t) defined for t0 ≤ t ≤ t1 .
The total angle encircled by the radius P C(t) as t
runs from t0 to t1 we will call the winding angle and
denote by αw :
y

y
C(t1 )

C(t1 )

P
αw
C(t0 )

P
αw

C(t)

C(t0 )

C(t)

x

x

This winding angle is insensitive to certain local
properties of the curve C(t) (e.g., local loops): in
the figures above, the winding angle is the same in
both cases (assuming the same points P , C(t0 ) and
C(t1 )).
The winding angle is positive if the point P lies
to the right with respect to the point traversing the
curve, and negative otherwise.

Of course, the absolute value of a winding angle
can be larger than 360◦ :
y
C(t1 )

exact length of the arc — an approximation suffices.
If Ba , Bb , Bc , and Bd are the points defining a Bézier
arc B (i.e., Ba and Bd are its nodes, Bb and Bc are
its control points), then
|Ba Bb | + |Bb Bc | + |Bc Bd | ≥ |B|

C(t0 )
P
αw >360◦

C(t)

x

For cyclic curves, the winding angle is always a multiple of 360◦ , i.e., αw = 360◦ w, where w is an integer. This entity w is called the winding number (for
a given point and curve).
w = +1

w = +1

w = +2

w = −1

In the following, we will focus our attention on cyclic
Bézier splines.
The idea of the algorithm computing the winding number for Bézier splines is due to Laurent
Siebenmann (metafont@ens.fr, 2000; now at http:
//lists.tug.org/metapost). Siebenmann’s solution, however, was MetaPost-oriented — it exploited
heavily the operation arctime, available in MetaPost
but unavailable in Metafont. Below, I’ll present an
algorithm based on the same idea but referring to
more elementary properties of a Bézier segment.
For a given point P and a Bézier spline C, we
will try to find the winding angle by measuring the
winding angles for a discrete series of time points.
First, we will try to measure the angles between
nodes 0, 1, 2, . . . , n of the spline C. If the relevant
Bézier segments are appropriately short, the sum
of the angles yields the total winding angle. The
problem arises when the Bézier arc is too long — see,
e.g., the leftmost panel of the first figure (the angle
C(t0 ) P C(t1 ) equals 360◦ − αw ).
The main observation of Siebenmann is as follows: if the length of the subarc C(t) for t0 ≤ t ≤ t1
is shorter than the length of the longer of the radii
P C(t0 ) and P C(t1 ), than we can safely assume that
the (acute) angle between P C(t0 ) and P C(t1 ) is the
winding angle. In fact, we do not need to know the

(with | . . . | denoting the length of an interval and the
length of a Bézier arc). In other words, we can safely
use the left-hand side of the above inequality instead
of the true value of the arc length in the computation
of the winding angle or winding number.
The algorithm can be expressed in pseudo-code
as follows:
input: a point P and a Bézier spline B,
consisting of segments B1 , B2 , . . . , Bn
output: αw — the winding angle for P and B
procedure windingangle(P, B)
if B is a single segment
let Ba , Bb , Bc , Bd be the consecutive
control nodes of the segment B
if min(|P Ba |, |P Bd |) < assumed minimal dist.
exit (P almost coincides with B,
winding angle incalculable)
fi
if |Ba Bb | + |Bb Bc | + |Bc Bd |
> max(|P Ba |, |P Bd |)
return windingangle(P, B(0, 1//2))
+windingangle(B(1//2, 1))
else
return angle α between the radii P Ba
and P Bd (−90◦ < α < 90◦ )
fi
else
return windingangle(P, B1 ) + . . .
+windingangle(P, Bn )
fi
end
A possible MetaPost/Metafont implementation:
% B is a B\’ezier segment
vardef mock_arclength(expr B) =
% |mock_arclength(B)>=arclength(B)|
length((postcontrol 0 of B)-(point 0 of B))
+ length((precontrol 1 of B)-(postcontrol 0 of B))
+ length((point 1 of B)-(precontrol 1 of B))
enddef;
% P is a point, B is a B\’ezier spline
vardef windingangle(expr P,B) =
if length(B)=1: % single segment
save r,v;
r0=length(P-point 0 of B);
r1=length(P-point 1 of B);
if (r0<2eps) and (r1<2eps):
% MF and MP are rather inaccurate, return 0
errhelp "Not advisable to continue.";
errmessage "windingangle: point almost "

& "coincides with B\’ezier segment (dist="
& decimal(min(r0,r1)) & ")";
0
else:
% v denotes both length and angle
v := mock_arclength(B);
% possibly too long B\’ezier arc?
if (v>r0) and (v>r1):
windingangle(P, subpath (0, 1/2) of B)
+ windingangle(P, subpath (1/2, 1) of B)
else:
v := angle((point 1 of B)-P) %difference
- angle((point 0 of B)-P);
if v >= 180: v := v-360; fi
%normalize
if v < -180: v := v+360; fi
v
%return
fi
fi
else: % multisegment spline
windingangle(P,subpath (0,1) of B)
for i:=1 upto length(B)-1:
+ windingangle(P,subpath (i,i+1) of B)
endfor
fi
enddef;

Although the returned angle (line marked ‘%return’
above) is acute, the difference of the component angles (lines at ‘%difference’) can be outside the interval h−180◦ , 180◦ i ; hence the normalization (lines
at %normalize).
If the operation windingnumber is needed for
some reason, it can be implemented trivially:

Postscriptum
In some cases, another definition, equivalent to the
one formulated above, may be useful (this formulation, given below without a proof of equivalence,
is a slightly edited excerpt from Siebenmann’s message):
Assume that we are given a curve C and point P .
Choose at random a line segment emanating from
the point P to the point W , with W outside the
bounding box of C and P . Inductively examine the
intersection points Q of P Q with C. Supposing these
points Q are all “nondegenerate” intersections, they
are also finite in number, and a sign +1 or −1 is
associated with each. Nondegenerate means that Q
is a smooth point of c and the tangent vector T to C
at Q is not parallel to P Q, and that Q is not a point
where C crosses itself. The sign to use is the sign of
the wedge product ‘(Q − P ) wedge T ’, i.e.,
(Q − P ) · (T rotated −90)
The sum of the signs is the winding number.
It is a probabilistic theorem that degenerate
intersections will rarely be met.
 Boguslaw Jackowski
Gdańsk, Poland
b_jackowski (at) gust dot org
dot pl

% P is a point, B is a B\’ezier spline
vardef windingnumber (expr P,B) =
windingangle(P,B)/360
enddef;

The operations windingangle or, equivalently, windingnumber can be used, e.g., for determining the
mutual position of two non-intersecting cyclic curves
(whether one is embedded inside the other or not):
tertiarydef a inside b =
if path a:
% |and path b|; |a| and |b| must be cyclic and
% must not touch each other
begingroup
save a_,b_;
(a_,b_) = (windingnumber(point 0 of a,b),
windingnumber(point 0 of b,a));
(abs(a_ - 1) < eps) and (abs(b_) < eps)
endgroup
else: % |numeric a and pair b|
begingroup
(a>=xpart b) and (a<=ypart b)
endgroup
fi
enddef;

